

GAMS による応用一般均衡分析： 資本市場と企業固有の生産要素

小 平 裕

1. はじめに
2. モデルの定式化
 - 2.1 貯蓄—投資の取扱い
 - 2.2 要素市場の取扱い
 - 2.3 均衡条件
3. モデルの方程式体系
4. ベンチマーク均衡データセット
5. GAMS 入力ファイル
6. 結果のまとめ

1. はじめに

本稿では、数理計画法解法プログラム GAMS を利用した応用一般均衡分析の練習として、基本モデル (age 1 モデル, 小平 (2002a)), 中間投入を考慮した拡張モデル (age 2 モデル, 小平 (2002b)) に続く age 3 モデルの構築を試みる。本稿の age 3 モデルは、前回の age 2 モデルを以下の 2 点について拡張している。第 1 に、資本市場をモデルに組み入れる。このことにより、貯蓄と投資をモデルの中で考察することが可能になる。第 2 に、企業固有の生産要素を考慮する。このことによって、要素価格の歪みを考察することが可能になる。

これまでの基本モデル age 1 や拡張モデル age2 では、要素の自由な移動を仮定しており、要素価格 (資本の賃貸率, 労働の賃金率) はその要素を使用する全ての企業で同一であるとされ、各企業は平均要素価格を払うと

考えられていた。しかし、現実の世界では、生産要素の移動が制限されて企業固有のものとなり、要素価格は企業毎に異なるという意味で、要素価格が歪みを持つ傾向がある。本稿の age 3 モデルでは、両要素について供給される固定量（与件）が完全雇用されるという設定を維持しながら、労働について企業固有の市場が成立する場合を考察する。すなわち、労働の賃金率は歪みを持つが、資本の賃貸率は歪みを持たない、つまり両企業を通じて同一である場合を取り上げる。以上が、第 2 の拡張である

age 2 モデルから本稿のモデルへの変更は上の 2 点に限られており、それを除けば全く同一である。すなわち、以下のような経済が想定される。2 種類の商品（CA と CB）と 2 種類の生産要素（労働 L と資本 K）があり、経済主体として 2 つの企業（FA と FB）と 2 つの家計（HA と HB）が存在する。企業は生産要素と商品の間投入を使い商品を生産して、財市場に供給する。ここで、生産に必要な生産要素は要素市場において家計から、中間需要は財市場において他の企業から購入する。企業は、自分の生産技術を制約として、収入と生産費用の差として定義される利潤が最大になるように生産計画（すなわち、要素需要、中間需要、商品供給）を立案する。家計は財市場において企業から商品を需要し消費する。家計は、要素市場において要素（労働と資本）を供給し、それから獲得する所得を制約として、消費の効用が最大になるように商品需要を計画する。何れの経済主体も価格受容者として行動する。また、モデルをオペレーショナルなものとするために、家計の効用関数および企業の生産関数は Cobb-Douglas 型の関数形で与えられるものと仮定する。

2. モデルの定式化

2.1 貯蓄－投資の取扱い

貯蓄と投資を考慮できるようにモデルを拡張するには、モデルに資本市場を組み入れる必要がある。その資本市場における需給バランスを検討す

際には、貯蓄と投資のどちらかが先に決定しており、もう一方がそれに合わせて決定されると考えるのが便利である。本稿では、(1)総貯蓄額が先に与えられ、それに見合うように総投資額が決定されると考える貯蓄主導型と、(2)総投資額が先に与えられ、それに見合うように総貯蓄額が決定されると考える投資主導型という2種類の貯蓄－投資の取扱いを検討する。

前者の貯蓄主導型では、

条件(a) 各家計の貯蓄割合 m_{ps}_h (限界貯蓄性向＝平均貯蓄性向) は、パラメータとして与えられている。

(b) 各家計はこの固定された貯蓄割合 m_{ps}_h で家計所得 YH_h から貯蓄に $m_{ps}_h YH_h$ を配分する。

(c) 家計部門の総貯蓄額は、家計貯蓄の合計 $\sum_{h \in H} m_{ps}_h YH_h$ として決定する。

(d) 家計の総貯蓄額に等しい総投資額 (=投資需要額) の投資が、企業部門によって行われる。

と想定する。ここで、商品 c の投資需要量を $QINV_c$ と表すと、企業部門の総投資額は $\sum_{c \in C} P_c QINV_c$ と表されるから、条件(d)はこの総投資額が家計部門の総貯蓄額 $\sum_{h \in H} m_{ps}_h YH_h$ に等しく決定されることを意味する。ただこれだけでは各商品への投資需要の大きさが決まらないので、ここではさらに

条件(e) 投資支出は、両財の投資需要量の比が一定となるように、2種類の商品の間に配分される。

と仮定する。商品 c の基準時点投資需要量を \overline{qinv}_c と表すと、これはパラメータとして与えられることになるので、条件(e)はある時点の投資需要量 $QINV_c$ が

$$QINV_c = IADJ \overline{qinv}_c$$

(投資方程式) により表されることを意味する。ただし、 $IADJ$ は比例的調整係数であり、両財に共通であると仮定される。以上より、条件(d)および

(e)の下では、総投資額や財価格が変化するとき、総投資額が与えられた総貯蓄額に等しくなるように、両財の投資需要量が比例的に調整されることになる。

逆に、投資主導型では、

条件(a) 企業部門の総投資額 (=投資需要額) はパラメーターとして与えられる。すなわち、投資調整係数 $IADJ$ の値は固定されている。

(b) これに見合う総貯蓄が、家計部門によりなされる。

と想定される。以上の条件だけでは必要な貯蓄がなされる保証はない。このことを保証するためには、家計貯蓄率の調整について何らかの規則を定めておく必要がある。両家計の貯蓄率が比例的に調整されるという想定も可能であるが、本稿では一方の家計（ここでは、家計 HA とする）の貯蓄率 mps_{HA} のみが調整され、残りの家計のそれは固定されていると想定する。すなわち、

条件(c) 家計 HA の貯蓄率 mps_{HA} は変数であり、その値は家計部門の総貯蓄額 $mps_{HA} YH_{HA} + mps_{HB} YH_{HB}$ が与えられた投資額 $\sum_{c \in C} P_c QINV_c$ に等しくなるように決定される。ただし、家計 HB の貯蓄率 mps_{HB} の値は固定されている。

と仮定する。

2.2 要素市場の取扱い

最初に述べたように、本稿のモデルでは要素価格の歪みを考慮する。すなわち、これまで基本モデルや age 2 モデルでは、全ての企業は両要素に同一の要素価格（資本の賃貸率、労働の賃金率）を支払うと仮定してきたのに対して、ここでは要素価格が企業毎に異なる可能性を考慮する。ただし、両要素のうち、要素価格に歪みが生じるのは労働のみとし、資本については賃貸率は両企業を通じて同一であると仮定する。また、両要素とも供給量（これは与件として固定されている）は全て雇用されるという完全雇用の

枠組みは、引き続き維持される。

具体的には、企業 a は要素 f について平均要素価格 WF_f の一定の割合 $wfdist_{fa} WF_f$ を支払うものとする。ここで、各要素の平均要素価格 WF_f は、当該要素の需給がバランスするように内生的に決定される。また、 $wfdist_{fa}$ は要素価格の歪み係数であり、外生的に与えられるパラメーターである。労働のみならず資本についても要素価格の歪みがある場合にも、この定式化は妥当する。本稿では資本については要素価格（貸貸率）の歪みはないと想定するので、

$$wfdist_{Ka} = 1$$

すなわち、資本要素価格の歪み係数は 1 に等しい。

2.3 均衡条件

このようなモデルの拡張により、均衡条件として財市場や要素市場のバランス条件（それぞれ複数）に加えて、資本市場のバランス条件が追加される。モデルを閉じるためには、これらの条件のうちの 1 つを落とすことも可能であるが、ここでは代わりに、貯蓄－投資バランス式にダミー変数 *WALRAS* を導入しよう。これにより、モデルの変数の数と方程式の数は等しくなる。ダミー変数 *WALRAS* は、一般均衡が成立する場合にそしてその場合に限り 0 という値をとり、貯蓄－投資バランスが成立する。

3. モデルの方程式体系

本稿の最終目的は、数理計画法解法プログラム GAMS を使用して応用一般均衡分析を行うことにあるので、GAMS のプログラミングを容易するために集合命令 SETS の利用を考えて、変数の添え字を表すのに集合を用いてモデルを構築する。本稿のモデルでも、以下の集合を使用する。

$$\begin{array}{ll} a \in A = \{FA, FB\} & \text{企業} = \{\text{企業 FA}, \text{企業 FB}\} \\ c \in C = \{CA, CB\} & \text{商品} = \{\text{商品 CA}, \text{商品 CB}\} \end{array}$$

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

$f \in F = \{K, L\}$ 生産要素 = {資本 K, 労働 L}

$h \in H = \{HA, HB\}$ 家計 = {家計 HA, 家計 HB}

さらに、GAMS は大文字、小文字を区別しないが、内生変数を表すには大文字を使い、パラメーター（固定値あるいは外生的な値をもつ変数を含む）を表すには小文字を使う、また数量を表す場合には変数あるいはパラメーターを Q あるいは q で始める、商品価格を表す場合には P あるいは p で始める、要素価格を表す場合には W あるいは w で始めるという小平 (2002a) の約束を、本稿でも引き続き維持することにする。パラメーターは次の通りである。

cpi = 消費者価格指数 CPI

$cwts_c$ = 消費者価格指数 CPI における商品 c の加重

ica_{ca} = 企業 a の産出 1 単位当たりの商品 c の中間投入量

mps_h = 家計 h の貯蓄割合（平均貯蓄性向 = 限界貯蓄性向）

qfs_f = 要素 f の供給量

\overline{qinv}_c = 商品 c の基準年投資需要量

$shry_{hf}$ = 要素 f の総要素所得のうち家計 h への分配率

$wfdist_{fa}$ = 企業 a における要素 f の要素価格歪み係数

α_{hc} = 家計 h の効用関数における商品 c の係数（商品 c への支出割合）

β_{fa} = 企業 a の生産関数における要素 f の係数（要素 f への分配率）

λ_a = 企業 a の生産関数の規模係数（効率性パラメーター）

θ_{ac} = 企業 a の活動水準 1 単位当たりの商品 c の産出量

モデルの方程式体系は次により与えられる。

(i) 生産および商品ブロック

Cobb-Douglas 型の生産技術を想定すると、企業 a の生産関数は、

$$(1) \quad QA_a = \lambda_a \prod_{f \in F} QF_{fa}^{\beta_{fa}} \quad a \in A$$

と表される。ただし、 QA_a は企業 a の産出量、 QF_{fa} は要素 f の投入量である。

要素価格の歪みも考慮するように拡張されたこのモデルでは、各企業の要素価格は経済全体の平均要素価格の一定の割合（歪み係数）になるとされているので、企業 a が雇用する要素 f に対して支払う要素価格は $wf_{dist_{fa}} WF_f$ となり、したがって雇用する要素 f 全体に対して支払う報酬は $wf_{dist_{fa}} WF_f QF_{fa}$ により与えられる。ここで想定しているように、Cobb-Douglas 型生産関数の場合には、生産要素 f への要素報酬は生産額 $PVA_a QA_a$ の一定割合 β_{fa} となることが知られているので、生産要素 f の要素（逆）需要関数は、

$$(2) \quad wf_{dist_{fa}} WF_f = \frac{\beta_{fa} PVA_a QA_a}{QF_{fa}} \quad f \in F, a \in A$$

により与えられる。age 2 モデルと比べると、この式は要素価格の歪みを導入したために修正されている。

本稿においても引き続き、固定係数（Leontief 技術）の中間投入を想定する。中間投入係数を ica_{ca} とすると、商品 c に対する企業 a の中間需要 $QINT_{ca}$ は、

$$(3) \quad QINT_{ca} = ica_{ca} QA_a \quad c \in C, a \in A$$

により与えられる。

企業 a の価格 PA_a は当該企業の産出 1 単位当たりの価値に等しいから

$$(4) \quad PA_a = \sum_{c \in C} \theta_{ac} P_c \quad a \in A$$

により与えられる。一方、企業 a の付加価値価格 PVA_a は企業価格 PA_a から産出 1 単位当たりの中間投入額 $\sum_{c \in C} P_c ica_{ca}$ を差し引いたものであるから、

$$(5) \quad PVA_a = PA_a - \sum_{c \in C} P_c ica_{ca} \quad a \in A$$

と表される。

商品 c の市場生産量は、両企業の商品 c の生産量 $\theta_{ac} QA_a$ の和となる。

$$(6) \quad Q_c = \sum_{a \in A} \theta_{ac} QA_a \quad c \in C$$

ただし、結合生産を想定していないので、企業 a の活動水準 1 単位当たりの商品 c の産出量を表す θ_{ac} の値は 0 または 1 である。

(ii) 制度ブロック

モデル拡張が家計行動に与える影響は、第 1 に要素価格が企業毎に異なることが家計 h の要素所得 YF_{hf} や総要素所得に与える影響を通じて、第 2 に貯蓄行動の導入が消費需要に与える影響を通じて表れる。

総要素所得は、基本モデルや age 2 モデルでは要素価格と総要素供給の積として表されたが、本稿のモデルでは第 1 の要因があるので簡単に積として表すことはできない。ここでは、どの企業がその要素を雇用するかを考慮する必要があるからである。企業 a による要素 f の雇用量を QF_{fa} とすると、この企業が支払う要素価格は $wf_{dist_{fa}} WF_f$ であるから、要素 f の総要素所得は $\sum_{a \in A} wf_{dist_{fa}} WF_f QF_{fa}$ と表される。したがって、家計 h が要素 f から受け取る要素所得 YF_{hf} は、

$$(7) \quad YF_{hf} = shry_{hf} \sum_{a \in A} wf_{dist_{fa}} WF_f QF_{fa} \quad h \in H, f \in F$$

により与えられる。ただし、 $shry_{hf}$ は要素 f の総要素所得のうち家計 h への分配率である。家計 h の総所得 YH_h は、資本および労働からの要素所得の和として与えられるから、

$$(8) \quad YH_h = \sum_{f \in F} YF_{hf} \quad h \in H$$

と表される。

次に、第2の要因である貯蓄行動を検討する。家計 h は (8) で与えられた家計所得 YH_h の一定割合 mps_h を貯蓄し、一定割合 $(1 - mps_h)$ を消費に支出する。すなわち、総消費支出額は $(1 - mps_h)YH_h$ である。また、効用関数が Cobb-Douglas 型の場合には、家計は総消費支出額の一定割合 α_{ch} を商品 c の購入に当てることが知られているから、家計 h の商品 c に対する (逆) 需要関数は、

$$(9) \quad QH_{ch} = \frac{\alpha_{ch}(1 - mps_h)YH_h}{P_c} \quad c \in C, h \in H$$

により与えられる。以上の式 (7) - (9) は全て、age 2 モデルと比べて修正されている。

家計部門の貯蓄により投資が行われる。貯蓄－投資行動が貯蓄主導型である場合には条件(e)により、投資主導型の場合には条件(a)により、投資支出は両財の投資需要量の比が一定となるように2種類の商品の間に配分されるので、投資方程式は

$$(10) \quad QINV_c = IADJ \overline{qinv}_c \quad c \in C$$

により与えられる。この式 (10) は本稿のモデル拡張により新たに追加される式である。

(iii) 体系制約ブロック

要素市場のバランス条件は、基本モデルや age 2 モデルと同様に

$$(11) \quad \sum_{a \in A} QF_{fa} = qfs_f \quad f \in F$$

により与えられる。

財市場のバランス条件にはモデル拡張の影響が見られる。財市場の供給側には変更はないが、需要側には家計による最終需要、企業による中間需

要の他に、投資需要が追加される。したがって、この市場のバランス条件は

$$(12) \quad Q_c = \sum_{h \in H} QH_{ch} + \sum_{a \in A} QINT_{ca} + QINV_c \quad c \in C$$

により与えられることになる。

貯蓄—投資行動をモデルに取り入れたために、本稿では資本市場のバランス条件が追加される。各市場のバランス条件のうちどれか1つ（例えば、資本市場のバランス条件）を消去することも可能であるが、そうする代わりに、ダミー変数 *WALRAS* を貯蓄—投資バランス式に導入して、全てのバランス条件を明示的に考察することにする。変数 *WALRAS* の値は一般均衡において0になる。

$$(13) \quad \sum_{c \in C} P_c QINV_c + WALRAS = \sum_{h \in H} mps_h YH_h$$

最後に、価格正規化方程式はこれまでと同じである。

$$(14) \quad \sum_{c \in C} cwtsc_c P_c = cpi$$

4. ベンチマーク均衡データセット

ベンチマーク均衡として、次のような社会勘定行列（表1）で与えられる数値例¹⁾を考えよう。同時に、仮設均衡として、資本ストックが10パーセント増加した場合を想定する。応用一般均衡分析の慣習に倣って、ベンチマーク均衡における商品価格は1に等しいものと想定するが、要素価格のうち労働の賃金率については歪みを考慮する。

表1に示されている社会勘定行列には、貯蓄—投資 *S-I* と名付けられた新しい勘定項目が含まれている。社会勘定行列は支払いの流れを示すだけなので、貯蓄主導型か投資主導型かという貯蓄—投資行動に関する想定の違いは、社会勘定行列には影響しない。

1) この数値例は、Lofgren (1999) による。

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

表 1 ベンチマーク均衡の社会勘定行列 (age 3 モデル)

		企業		商品		資本 労働		家計		貯蓄-投資	合計
		FA	FB	CA	CB	K	L	HA	HB	S-I	
企業	FA			250							250
	FB				305						305
商品	CA	60	40					50	75	25	250
	CB	40	60					100	50	55	305
資本 労働	K	78	125								203
	L	72	80								152
家計	HA					120	80				200
	HB					83	72				155
貯蓄-投資	S-I							50	30		80
合計		250	305	250	305	203	152	200	155	80	

貯蓄-投資 S-I の行は、家計（この経済における唯一の貯蓄者）からの支払の受け取りを示している。貯蓄-投資 S-I の行を左から右に読み進むと、ベンチマーク均衡では家計 HA は 50 の、家計 HB は 30 の貯蓄をしており、家計部門の貯蓄合計は 80（いずれも金額）であることが分かる。貯蓄-投資 S-I の列は、投資に利用される商品への支出を示している。このモデルでは投資は、自社の資本ストック追加分として投資財を受け取る仕向け先の企業ではなく、資本ストックの生産に利用される商品の由来によって定義されている。貯蓄-投資 S-I の列を上から下に読むと、ベンチマーク均衡では投資需要として商品 CA が 25、商品 CB が 55 購入されており、合計 80（いずれも金額、ただしベンチマーク均衡では両商品の財価格は 1 とされているので、これらの数値は数量を表していると考えられることも可能である）の投資が行われていることが分かる。

なお、投資は購入される商品の由来によって定義されていることから、このモデルが当てはまるのは、新たな投資が生産能力の拡大をもたらさない程度の非常に短い期間に限られることになる。より長い期間を考察する

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

モデル（例えば、多期間モデル）では、資本ストックの増加による生産能力の拡大を明示的に考察することが必要になる。今後の課題である。

本稿のモデルでは、要素市場のうち労働について要素価格の歪みを考えている。雇用される労働量は企業 FA で 100 単位、企業 FB で 50 単位である。資本については両企業間の自由な移動を認めているので、両企業に共通な資本貸付率（要素価格）のベンチマーク均衡値が 1 であることを使って、資本の要素需要量を求めている。生産要素の取扱いが age 2 モデルから age 3 モデルへこのように変更されたが、社会勘定行列に変更はない。

5. GAMS 入力ファイル

上の 2 節および 3 節で検討した age 3 モデルの方程式体系と、4 節のベンチマーク均衡データセットに基づく一般均衡体系を、数理計画法解法プログラム GAMS のプログラムとして表現したものが、表 2 の入力ファイルである。表の左側の数字は説明のために付けた行番号であり、GAMS 入力には無関係である。行番号に*（アスタリスク）が付いている行は、age 2 モデルと異なる箇所である。この入力ファイルでは、最初にベンチマーク均衡の再現テストを行い、続いて資本ストックが 10 パーセント増加した場合の仮設均衡を求める。

本稿では、資本市場の組み入れと要素価格の歪みの 2 点について age 2 モデル（小平 (2002b)）の拡張を試みた。第 1 の拡張はモデルの中で貯蓄と投資を考慮することを可能にするが、このことがどのような GAMS コードの変更を必要とするか、入力ファイルの該当箇所を見ていこう。

社会勘定行列に関しては、勘定項目に新たに貯蓄－投資勘定 S-I が加わるので、そのための修正が必要である。具体的には、集合の定義（15 行以下）の中で勘定項目を表す大域集合 AC の要素に、新たに貯蓄－投資勘定 S-I を追加した（28 行）うえで、社会勘定行列 SAM を定義する（162-176 行）。

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

表 2 モデル age 2 の入力ファイル

```

*1 $TITLE AGE3
2 $OFFUPPER
3 *$ONSYMXREF
4
5 *INTRODUCTION=====
6 $ONTEXT
7
8 AGE3 = Extended Model with
9         (1) savings and investment
10        (See CLOSURE FOR SAVINGS-INVESTMENT BALANCE)
11        (2) distortion in labor wages
12
13 $OFFTEXT
14
15 *SETS=====
16
17 SETS
18
19 AC global set (SAM accounts and other items)
20   /FA   firm fa
21   FB   firm fb
22   CA   commodity ca
23   CB   commodity cb
24   K    capital
25   L    labor
26   HA   household ha
27   HB   household hb
*28   S-I  savings-investment
29   TOTAL total account in SAM /
30
31 ACNT(AC) all elements in AC except total
32
33 A(AC) firms
34     /FA, FB/
35
36 C(AC) commodities
37     /CA, CB/
38
39 F(AC) factors
40     /K, L/
41
42 H(AC) households
43     /HA, HB/
44 ;
45
46 ALIAS(AC,ACP); ALIAS(C,CP); ALIAS(F,FP);

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

47 ACNT(AC) = YES; ACNT('TOTAL') = NO; ALIAS(ACNT,ACNTP);
48
49
50 *PARAMETERS=====
51
52 PARAMETERS
53
54 cpi          consumer price index
55 cwts(C)     weight of commodity c in the CPI
56 ica(C,A)   firm a's intermediate input of commodity c
57 qfs(F)     supply of factor f
*58 qinvbar(C) base-year investment demand for commodity c
59 shry(H,F)  household h's share in factor f's income
*60 wfdist(F,A) distortion factor for factor f in firm a
61
62 alpha(C,H) commodity c's share in household h's spending
63 beta(F,A)  factor f's share in firm a's value-added
64 lambda(A)  scale parameter for firm a's production
65 theta(A,C) commodity c's output per unit of production by firm a
66 ;
67
68 *VARIABLES=====
69
70 VARIABLES
71
*72 IADJ      investment adjustment coefficient
*73 MPS(H)   mps (=aps) for household h
74 P(C)      price of commodity c
75 PA(A)     price of firm a
76 PVA(A)   value-added (or net) price for firm a
77 Q(C)     output level for commodity c
78 QA(A)   production level of firm a
79 QF(F,A) firm a's demand of factor f
80 QH(C,H) household h's demand of commodity c
81 QINT(C,A) firm a's intermediate demand of commodity c
*82 QINV(C) investment demand for commodity c
*83 WALRAS   dummy variable (zero at equilibrium)
84 WF(F)    average price of factor f
85 YF(H,F)  household h's income from factor f
86 YH(H)    income of household h
87 ;
88
89 *EQUATIONS=====
90
91 EQUATIONS
92
93 *PRODUCTION AND COMMODITY BLOCK+++++++
94 PRODFN(A)  firm a's production function

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

95  FACDEM(F,A)      firm a's demand for factor f
96  INTDEM(C,A)     firm a's intermediate demand for commodity c
97  OUTPUTFN(C)     output of commodity c
98  PADEF(A)        price for firm a
99  PVADEF(A)       value-added price for firm a
100
101 *INSTITUTION BLOCK+++++
102 FACTTRNS(H,F)   transfer from factor f to household h
103 HHDINC(H)       income of household h
104 HHDEM(C,H)     household h's demand of commodity c
*105 INVDEM(C)      investment demand for commodity c
106
107 *SYSTEM CONSTRAINT BLOCK+++++
108 FACTEQ(F)       market balance for factor f
109 COMEQ(C)       market balance for commodity c
*110 SAVINV        savings-investment balance
111 PNORM          price normalization
112 ;
113
114 *PRODUCTION AND COMMODITY BLOCK+++++
115
116 PRODFN(A)..     QA(A) =E= lambda(A)*PROD(F, QF(F,A)**beta(F,A));
117
*118 FACDEM(F,A).. WF(F)*wfdist(F,A)
119                =E= beta(F,A)*PVA(A)*QA(A) / QF(F,A);
120
121 INTDEM(C,A)..   QINT(C,A) =E= ica(C,A)*QA(A);
122
123 OUTPUTFN(C)..   Q(C) =E= SUM(A, theta(A,C)*QA(A));
124
125 PADEF(A)..      PA(A) =E= SUM(C, theta(A,C)*P(C));
126
127 PVADEF(A)..     PVA(A) =E= PA(A) - SUM(C, P(C)*ica(C,A));
128
129
130 *INSTITUTION BLOCK+++++
131
*132 FACTTRNS(H,F).. YF(H,F)
133                =E= shry(H,F)*SUM(A, WF(F)*wfdist(F,A)*QF(F,A));
134
135 HHDINC(H)..     YH(H) =E= SUM(F, YF(H,F));
136
*137 HHDEM(C,H)..   QH(C,H) =E= alpha(C,H)*(1 - MPS(H))*YH(H)/P(C);
138
*139 INVDEM(C)..    QINV(C) =E= qinvbar(C)*IADJ;
140
141
142 *SYSTEM CONSTRAINT BLOCK+++++

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

143
144  FACTEQ(F)..      SUM(A, QF(F,A)) =E= qfs(F);
145
*146  COMEQ(C)..      Q(C) =E=
147                  SUM(H, QH(C,H)) + SUM(A, QINT(C,A)) + QINV(C);
148
*149  SAVINV..        SUM(C, P(C)*QINV(C)) + WALRAS
150                  =E= SUM(H, MPS(H)*YH(H));
151
152  PNORM..          SUM(C, cwts(C)*P(C)) =E= cpi;
153
154
155  *MODEL=====
156
157  MODEL
*158  AGE3 Model with Savings-Investment and Wage Distortions
159  /ALL/
160  ;
161
162  *SOCIAL ACCOUNTING MATRIX=====
163
164  TABLE SAM(AC,ACP)  social accounting matrix
165
*166      FA  FB  CA  CB  K  L  HA  HB  S-I
167  FA              250
168  FB              305
169  CA      60    40
170  CB      40    60
171  K       78   125
172  L       72    80
173  HA              120  80
174  HB              83   72
*175  S-I              50  30
176  ;
177
178
179  PARAMETER
180  tdiff(AC)  column minus row total for account ac;
181  *This parameter is used to check that the above SAM is balanced.
182  SAM('TOTAL',ACNTP) = SUM(ACNT, SAM(ACNT,ACNTP));
183  SAM(ACNT,'TOTAL') = SUM(ACNTP, SAM(ACNT,ACNTP));
184  tdiff(ACNT)       = SAM('TOTAL',ACNT) - SAM(ACNT,'TOTAL');
185
186  DISPLAY SAM, tdiff;
187
188
189  *ASSIGNMENTS FOR PARAMETERS AND VARIABLES=====
190

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

191 PARAMETERS
192 *The following parameters are used to define initial values of
193 *model variables.
*194 IADJO, MPSO(H),
195 P0(C), PA0(A), PVA0(A),
*196 Q0(C), QA0(A), QF0(F,A), QH0(C,H), QINT0(C,A), QINV0(C),
197 WF0(F), YF0(H,F), YH0(H)
198 ;
199
200
*201 *FACTOR EMPLOYMENT AND PRICES+++++++
*202 *This section is new and includes all items related to factor
*203 *prices and quantities.
*204
*205 PARAMETERS
*206 labor(A)      no. of workers employed by firm a
*207              /FA 100, FB 50/
*208
*209 wfa(F,A)      calibrated factor f's cost in firm a
*210 cdiff(F,A)    difference btwn wfa and SAM value
*211 ;
212
213 *Defining factor employment and supply
*214 QF0('L',A) = labor(A);
215 QF0('K',A) = SAM('K',A);
216 qfs(F)     = SUM(A, QF0(F,A));
217
*218 *Computing firm-specific wage
*219 wfa(F,A)   = SAM(F,A)/QF0(F,A);
220
221 *Computing average wage
222 WF0(F)     = SUM(A, SAM(F,A))/SUM(A, QF0(F,A));
223
*224 *Computing wage distortion factors
*225 wfdist(F,A) = wfa(F,A) / WF0(F);
226
*227 *Checking calibration
*228 cdiff(F,A) = WF0(F)*wfdist(F,A)*QF0(F,A) - SAM(F,A);
229
*230 DISPLAY wfa, wfdist;
231
232
233 *PRODUCTION AND COMMODITY BLOCK+++++++
234
235 P0(C)      = 1;
236 PA0(A)     = 1;
237 PVA0(A)    = SUM(F, SAM(F,A)) / (SAM(A, 'TOTAL')/PA0(A));
238

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

239  Q0(C)      = SAM('TOTAL',C)/P0(C);
240  QA0(A)     = SAM('TOTAL',A)/PA0(A);
*241  QINT0(C,A) = SAM(C,A)/P0(C);
242
243  ica(C,A)   = (SAM(C,A)/P0(C)) / QA0(A);
244  beta(F,A) = SAM(F,A) / SUM(FP, SAM(FP,A));
245  lambda(A)  = QA0(A) / PROD(F, QF0(F,A)**beta(F,A));
246  theta(A,C) = (SAM(A,C)/P0(C)) / QA0(A);
247
248
249  *INSTITUTION BLOCK+++++
250
*251  IADJ0     = 1;
*252  MPS0(H)   = SAM('S-I',H)/SAM('TOTAL',H);
253  QH0(C,H)   = SAM(C,H)/P0(C);
*254  QINV0(C)  = SAM(C,'S-I')/P0(C);
255  YF0(H,F)   = SAM(H,F);
256  YH0(H)     = SAM('TOTAL',H);
257
258  shry(H,F)  = SAM(H,F)/SAM('TOTAL',F);
259  alpha(C,H) = SAM(C,H)/SUM(CP, SAM(CP,H));
*260  qinvbar(C) = SAM(C,'S-I')/P0(C);
261
262  *SYSTEM CONSTRAINT BLOCK+++++
263
264  cwtS(C)    = SUM(H, SAM(C,H)) / SUM((CP,H), SAM(CP,H));
265  cpi        = SUM(C, cwtS(C)*P0(C));
266
267
268  *INITIALIZING ALL VARIABLES+++++
269
*270  IADJ.L     = IADJ0;
*271  MPS.L(H)   = MPS0(H);
272  P.L(C)      = P0(C);
273  PA.L(A)     = PA0(A);
274  PVA.L(A)    = PVA0(A);
275  Q.L(C)      = Q0(C);
276  QA.L(A)     = QA0(A);
277  QF.L(F,A)   = QF0(F,A);
278  QH.L(C,H)   = QH0(C,H);
279  QINT.L(C,A) = QINT0(C,A);
*280  QINV.L(C)  = QINV0(C);
281  WF.L(F)     = WFO(F);
282  YF.L(H,F)   = YF0(H,F);
283  YH.L(H)     = YH0(H);
284  ;
285
286

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

287 *DISPLAY+*****
288
289 DISPLAY
*290 cpi, cwts, ica, qfs, qinvbar, shry, wfdist,
291 alpha, beta, lambda, theta,
*292 IADJ.L, MPS.L, P.L, PA.L, PVA.L,
*293 Q.L, QA.L, QF.L, QH.L, QINT.L, QINV.L,
294 WF.L, YF.L, YH.L
295 ;
296
297
*298 *CLOSURE FOR SAVINGS-INVESTMENT BALANCE
*299 *savings-driven investment
*300 MPS.FX(H) = MPS0(H);
*301
*302 *investment-driven savings
*303 * IADJ.FX = IADJ0;
*304 * MPS.LO('HA') = -INF;
*305 * MPS.UP('HA') = +INF;
*306 * MPS.L('HA') = MPS0('HA');
*307 * MPS.FX('HB') = MPS0('HB');
308
309 *SOLVE STATEMENT FOR BASE=====
310
311 AGE3.HOLDFIXED = 1;
312
313 *SOLVE AGE3 USING MCP;
314
315
316 *REPORT SETUP AND BASE REPORT=====
317
318 *SET AND PARAMETERS FOR REPORTS+++++++
319
320 SET
321 SIM simulations
322 /BASE base simulation
323 HYPO 10 per cent increase in capital stock/
324 ;
325
326 PARAMETERS
327
328 QFSKSIM(SIM) capital supply for simulation (experiment parameter)
329 *Parameter is used to change the value for the capital stock parameter
330 *before solving the model for simulation sim
331
332 QFSREP(F,SIM) supply of factor f for simulation sim (value used)
*333 IADJREP(SIM) investment adjustment factor
*334 MPSREP(H,SIM) mps (=aps) for household h

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

335 PREP(C,SIM)      demand price for commodity c
336 PAREP(A,SIM)    price of firm a
337 PVAREP(A,SIM)  value-added price for firm a
338 QREP(C,SIM)     output level for commodity c
339 QAREP(A,SIM)    production level of firm a
340 QFREP(F,A,SIM)  firm a's demand of factor f
341 QHREP(C,H,SIM)  household h's demand of commodity c
342 QINTREP(C,A,SIM) firm a's intermediate input of commodity c
*343 QINVREP(C,SIM) quantity of investment by commodity of origin c
*344 WALRASREP(SIM) dummy variable (zero at equilibrium)
345 WFREP(F,SIM)    average price of factor f
*346 WFAREP(F,A,SIM) price of factor f for firm a
347 YFREP(H,F,SIM) household h's income from factor f
348 YHREP(H,SIM)    income of household h
349 SAMREP(SIM,AC,ACP) SAM computed from model solution
350 SAMCHK(AC,SIM)  column minus row total for account ac in SAM
351 ;
352
353
354 QFSKSIM('BASE') = qfs('K');
355
356 *10 per cent increase in capital stock
357 QFSKSIM('HYPO') = 1.1*qfs('K');
358
359 DISPLAY QFSKSIM;
360
361
362 LOOP(SIM,
363
364   qfs('K') = QFSKSIM(SIM);
365
366 SOLVE AGE3 USING MCP;
367
368 QFSREP(F,SIM) = qfs(F);
369
*370 MPSREP(H,SIM) = MPS.L(H);
371 IADJREP(SIM)   = IADJ.L;
372 PREP(C,SIM)    = P.L(C);
373 PAREP(A,SIM)   = PA.L(A);
374 PVAREP(A,SIM) = PVA.L(A);
375 QREP(C,SIM)    = Q.L(C);
376 QAREP(A,SIM)   = QA.L(A);
377 QFREP(F,A,SIM) = QF.L(F,A);
378 QHREP(C,H,SIM) = QH.L(C,H);
379 QINTREP(C,A,SIM) = QINT.L(C,A);
*380 QINVREP(C,SIM) = QINV.L(C);
*381 WALRASREP(SIM) = WALRAS.L;
382 WFREP(F,SIM)   = WF.L(F);

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

```

*383  WFAREP(F,A,SIM) = WF.L(F)*wfdist(F,A);
384  YFREP(H,F,SIM) = YF.L(H,F);
385  YHREP(H,SIM) = YH.L(H);
386
387  *SAM items
388  *Factor payments
*389  SAMREP(SIM,F,A) = wfdist(F,A)*WF.L(F)*QF.L(F,A);
390  *Payments from commodities
391  SAMREP(SIM,A,C) = P.L(C)*theta(A,C)*QA.L(A);
392  *Payments from factors
393  SAMREP(SIM,H,F) = YF.L(H,F);
394  *Household demands
395  SAMREP(SIM,C,H) = P.L(C)*QH.L(C,H);
396  *Intermediate demands
397  SAMREP(SIM,C,A) = P.L(C)*ica(C,A)*QA.L(A);
*398  *Savings supplies
*399  SAMREP(SIM,'S-I',H) = MPS.L(H)*YH.L(H);
*400  *Investment demands
*401  SAMREP(SIM,C,'S-I') = P.L(C)*IADJ.L*qinvbar(C);
402
403  );
404
405  *Computing totals for SAMREP
406  SAMREP(SIM,'TOTAL',ACNTP) = SUM(ACNT, SAMREP(SIM,ACNT,ACNTP));
407  SAMREP(SIM,ACNT,'TOTAL') = SUM(ACNTP, SAMREP(SIM,ACNT,ACNTP));
408
409  *Check that SAMREP is balanced
410  SAMCHK(ACNT,SIM) = SAMREP(SIM,'TOTAL',ACNT) - SAMREP(SIM,ACNT,'TOTAL')
411
412  OPTION QFREP:3:1:1, QHREP:3:1:1, YFREP:3:1:1, SAMREP:3:1:1;
413
414  DISPLAY
415  IADJREP, MPSREP,
416  PREP, PAREP, PVAREP,
417  QREP, QAREP, QFREP, QFSREP, QHREP, QINTREP, QINVREP,
418  WALRASREP, WFREP, WFAREP, YFREP, YHREP,
419  SAMREP, SAMCHK
420  ;

```

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

家計の意思決定は、最初に貯蓄するか今期消費するか、次に今期の消費支出を各商品へどう振り分けるかという2段階でなされると想定したので、商品に対する（逆）需要関数(9)は、貯蓄に関する意思決定の影響を受ける(137行)。財市場のバランス条件(12)においても、需要側に投資需要が追加されるというモデル拡張の影響がある(146-147行)。投資方程式(10)と資本市場のバランス条件(13)は、モデルの拡張により age 3 モデルで初めて加えられた(139行, 149-150行)。

資本市場における貯蓄—投資決定の様子については、2節で検討したように貯蓄主導型と投資主導型の2通りの説明が可能である。この部分は入力ファイルの298行以下でプログラム化されている。貯蓄主導型では、貯蓄率 mps_h が初期値に固定され、家計部門が行った貯蓄に等しい投資がなされると想定するが、その GAMS コードは入力ファイル 299-300 行である。GAMS では変数のレコードは、

- . LO = 下限
- . L = 主問題の解
- . UP = 上限
- . M = 双対問題の解

の4つのフィールドを持つが、300行の変数 MPS につけられた拡張子 FX は、その変数の上限と下限を右辺の値に等しく設定する機能を持つ。したがって、300行の

$$MPS.FX(H) = MPS0(H);$$

は正しく、貯蓄率を初期値に固定することに他ならない。

他方、投資主導型では、投資調整計数 $IDAJ$ が初期値に固定され、必要な貯蓄が家計部門でなされるように家計 HA の貯蓄率が調節されると想定するが、その GAMS コードは 302-307 行である。すなわち、303 行の

$$IADJ.FX = IADJ0;$$

で投資調整計数を初期値に固定し、307 行の

$$\text{MPS.L('HB')} = \text{MPS0('HB')};$$

で家計 **HB** の貯蓄率を初期値に設定する。調節される家計 **HA** の貯蓄率については、304-306 行の

$$\text{MPS.LO('HA')} = -\text{INF};$$

$$\text{MPS.UP('HA')} = +\text{INF};$$

$$\text{MPS.L('HA')} = \text{MPS0('HA')};$$

で下限を $-\infty$ 、上限を $+\infty$ として、とりうる値について制限を外して計算している。表 2 では 302-307 行の行頭に* (アスタリスク) が付けられ、この部分は注釈文になっているので、表 2 の入力ファイルをそのまま実行して得られる計算結果は、貯蓄主導型を想定した場合の結果になる。投資主導型の結果を得るには、300 行を注釈文にし 303-307 行の*を外して入力ファイルを実行すればよい。

第 2 の拡張は、要素価格の歪みをモデルに組み入れることであった。ただし、歪みがあるのは労働の賃金率だけで、資本の賃貸率は歪みを持たないと想定された。しかし、本稿の理論モデルは、企業 a が雇用する要素 f の歪み係数 $wf_{dist_{fa}}$ を定義した上で、歪みを持たない資本については $wf_{dist_{ka}} = 1$ として、この拡張を処理しているので、2 節でも指摘したように、どちらの(あるいは両方の)要素に歪みがあっても本稿の理論モデルは妥当する。入力ファイルの GAMS コードもそうである。要素価格の歪みについては、入力ファイル 201-230 行で処理している。

- (i) 企業別の要素需要量 QF_{fa} の初期値を、ベンチマーク・データセットから求める (215 行)。要素供給量 (外生的) qfs_f の初期値は、206-207 行で与えられ、214 行で代入される。
- (ii) カラプレイションを容易にするために、企業別の要素価格 wfa_{fa} を新たに導入し (209 行)、初期値を計算する (218-219 行)。経済全体の平均要素価格 WF_f の初期値は、ベンチマーク均衡の社会勘定行列の要素所

得値と要素需要量 QF_{fa} の初期値から算出する (221-222 行)。

(iii) 企業別要素価格 wfa_{fa} の初期値と平均要素価格 WF_f の初期値の比として定義される要素価格歪み係数 $wfdist_{fa} = \frac{wfa_{fa}}{WF_f}$ を計算する (224-225 行)。

(iv) 各企業について要素毎に、歪み係数と平均要素価格と要素需要量の積 $wfdist_{fa} WF_f QF_{fa}$ が、社会勘定行列における当該企業による当該要素への支払いに等しいを確認する。このために、入力ファイルでは `cdiff` というパラメーターを定義し (210 行)、計算している (227-228 行)。

309 行以下が解法である。ここでは、ベンチマーク均衡 **BASE** と仮設均衡 **HYPO** (資本ストックが 10% 増加した場合) の 2 つの均衡解を、**LOOP** 命令を利用して 1 つの入力コードで求めようとしている。そのために、320-324 行の **SET** 命令で **LOOP** を回すための添え字を、また 326-351 行の **PARAMETERS** 命令で変数を準備する。**LOOP** 文は 362 行から 403 行にわたっており、途中の 366 行に **SOLVE** 命令がある。**LOOP** 文の中の 387-401 行は社会勘定行列を求めるための計算であり、405-410 行は求めた社会勘定行列がバランスしているかを (行和と列和の差がほぼ 0 になるかどうかで) 確認するための計算である。414-420 行の **DISPLAY** 命令は、ベンチマーク均衡と仮設均衡の均衡値の比較を出力する。

6. 結果のまとめ

前節の入力ファイル (表 2) を数理計画法解法プログラム **GAMS** でコンパイルし実行すると、出力ファイルが得られる。出力ファイルは大変長いのでここには掲載しないが、注意すべき箇所について出力ファイルの見方を説明しよう。

出力ファイルの最初の部分は、エコープリントと呼ばれる入力ファイルのオウム返しの出力である。エコープリントが終わると、**GAMS** の処理結果が記されている。最初に注目する箇所は、ベンチマーク均衡 (入力フ

ファイル 162-176 行) の入力チェックの処理結果である。ここでは、データ入力は社会勘定行列の形でなされているが、社会勘定行列の列和と行和は等しいという性質に注目して、列和と行和の差 `tdiff` を定義し、これらが全て 0 になるかどうかで、正しく入力されたかどうかを判定しようとしている。入力ファイル 186 行の「`tdiff` を示せ」という命令に対する結果は「全て 0」となっており、誤りなく入力されたことが確認される。

次に注目する箇所は、入力ファイル 201-230 行で行った企業毎の要素価格と要素価格の歪み係数のカラブレイションと、入力ファイル 233-296 行の財価格以外のパラメーターのカラブレイションの結果である (表 3 参照)。ここには変数の初期値 (ベンチマーク均衡値) も出力されており、後に示されている再現テストの変数値とこれらの値が一致することを確認する必要がある。

貯蓄—投資の扱いについて、貯蓄主導型か投資主導型か何れかの指定をした (入力ファイル 298-307 行) 後、入力ファイルの `LOOP` 文 (281-301 行) の途中の `SOLVE` 命令 (285 行) に対する状況報告が出力されている。ここでは、`SOLVER STATUS` (解法プログラムの状態) が `NORMAL COMPLETION` (正常終了) となっており、`MODEL STATUS` (モデルの状態) が `OPTIMAL` (最適解を見付けて終了) となっていることを確認することが重要である。

`LOOP` 文の 1 巡目 `BASE` (ベンチマーク均衡) は、パラメーターにベンチマーク値を与えた場合の均衡値の計算、すなわち再現テストである。表 4 にまとめた出力ファイルの結果を見ると、再現テストに合格していることが分かり、モデルの定式化に成功したと判断される。2 巡目 `HYPO` (仮設均衡) は、資本ストックが 10% 増加した場合の均衡 (仮設均衡) の計算である。入力ファイルの 419 行で出力される仮設均衡の社会勘定行列を表 5 にまとめた。社会勘定行列の数値は数量を示すものではなく、取引金額すなわち価格×数量であることに注意して欲しい。

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

計算は、貯蓄主導型と投資主導型の場合に分けて2回を行った。均衡値や社会勘定行列を表の形にまとめておく。

表3 カラフレイトされたパラメーター値 (age 3 モデル)

cpj	消費者価格指数 CPI 1.000	wfdist (F,A)	企業 a における要素 f の 歪み係数
cwts (C)	消費者価格指数 CPI にお ける商品の加重	(K,FA)	1.000
CA	0.455	(K,FB)	1.000
CB	0.545	(L,FA)	0.711
		(L,FB)	1.579
ica (C,A)	企業 a の産出単位当たり の商品 c の中間投入量	alpha (H,C)	家計の効用関数における 商品 c の係数 (商品 c への支出割合)
(CA,FA)	0.240	(HA,CA)	0.333
(CB,FA)	0.160	(HA,CB)	0.667
(CA,FB)	0.131	(HB,CA)	0.600
(CB,FB)	0.197	(HB,CB)	0.400
qfs (F)	要素 f の供給量	beta (F,A)	企業 a の生産関数におけ る要素 f の係数 (企業 a の付加価値に占 める要素 f の割合)
K	203.000	(K,FA)	0.520
L	150.000	(L,FA)	0.480
qinvbar (C)	商品 c の投資需要 (初期値)	(K,FB)	0.610
CA	25.000	(L,FB)	0.390
CB	55.000	lambda (A)	企業 a の生産関数の規模 係数
shry (H,F)	要素 fk 総要素所得のう ち家計 h への分配率	FA	2.845
(HA,K)	0.591	FB	3.489
(HB,K)	0.409	theta (A,C)	企業 a の活動水準単位当 たりの商品 c の産出量
(HA,L)	0.526	(FA,CA)	1.000
(HB,L)	0.474	(FA,CB)	0.000
wfa (F,A)	企業における要素のカラ フレイトされた費用	(FB,CA)	0.000
(K,FA)	1.000	(FB,CB)	1.000
(K,FB)	1.000		
(L,FA)	0.720		
(L,FB)	1.600		

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

表 4 均衡の比較 (age 3 モデル)

	ベンチ マーク均衡	仮 設 均 衡	
		貯蓄主導	投資主導
商品価格 P			
商品 CA	1.000	1.003	1.003
CB	1.000	0.997	0.997
企業価格 PA			
企業 FA	1.000	1.003	1.003
FB	1.000	0.997	0.997
企業付加価値価格 PVA			
企業 FA	0.600	0.603	0.603
FB	0.672	0.670	0.670
産出量 Q			
商品 CA	250.000	263.388	263.473
CB	305.000	322.382	322.275
活動水準 QA			
企業 FA	250.000	263.388	263.473
FB	305.000	322.382	322.275
要素需要 QF			
企業 FA 資本 K	78.000	86.087	86.123
労働 L	100.000	100.181	100.203
企業 FB 資本 K	125.000	137.213	137.177
労働 L	50.000	49.819	49.797
家計需要 QH			
家計 HA 商品 CA	50.000	52.595	54.083
CB	100.000	105.818	108.813
家計 HB 商品 CA	75.000	78.895	78.890
CB	50.000	52.910	52.908
中間需要 QINT			
企業 FA 商品 CA	60.000	63.213	63.233
CB	40.000	42.142	42.156
企業 FB 商品 CA	40.000	42.280	42.266
CB	60.000	63.419	63.398

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

	ベンチ マーク均衡	仮 設 均 衡	
		貯蓄主導	投資主導
投資需要 QINV			
商品 CA	25.000	26.405	25.000
CB	55.000	58.092	55.000
平均要素価格 WF			
資本 K	1.000	0.959	0.959
労働 L	1.013	1.071	1.071
企業別要素価格 WFA			
企業 FA 資本 K	1.000	0.959	0.959
労働 L	0.720	0.761	0.761
企業 FB 資本 K	1.000	0.959	0.959
労働 L	1.600	1.691	1.691
家計要素所得 YF			
家計 HA 資本 K	120.000	126.613	126.603
労働 L	80.000	84.450	84.448
家計 HB 資本 K	83.000	87.574	87.567
労働 L	72.000	76.005	76.003
家計所得 YH			
家計 HA	200.000	211.063	211.051
家計 HB	155.000	163.579	163.570
限界貯蓄性向 MPS			
家計 HA	0.250	0.250	0.229
家計 HB	0.194	0.194	0.194
投資調整係数 IADJ	1.000	1.056	1.000

表5 仮設均衡の社会勘定行列 (age 3 モデル)

(1) 貯蓄主導型	企業		商品		資本		労働		家計		貯蓄投資		合計
	FA	FB	CA	CB	K	L	HA	HB	S-I				
企業 FA			264.243										264.243
企業 FB				321.510									321.510
商品 CA	63.418	42.417					52.766	79.151	26.491				264.243
商品 CB	42.028	63.248					105.531	52.767	57.935				321.509
資本 K	82.574	131.613											214.187
労働 L	76.222	84.232											160.454
家計 HA					126.613	84.450							211.063
家計 HB					87.574	76.005	52.766	31.660					163.579
貯蓄投資 S-I													84.426
合計	264.242	321.510	264.243	321.510	214.187	160.455	211.063	163.578	84.426				
列和と行和の差	-0.001	0.000	0.000	0.001	0.000	0.001	0.000	-0.001	0.000				0.000
(2) 投資主導型													
企業 FA			264.329										264.329
企業 FB				321.402									321.402
商品 CA	63.439	42.403					54.259	79.147	25.081				264.329
商品 CB	42.041	63.227					108.518	52.765	54.851				321.402
資本 K	82.601	131.568											214.169
労働 L	76.247	84.204											160.451
家計 HA					126.603	84.448							211.051
家計 HB					87.567	76.003	48.274	31.659					163.570
貯蓄投資 S-I													79.933
合計	264.328	321.402	264.329	321.402	214.170	160.451	211.051	163.571	79.932				
列和と行和の差	-0.001	0.000	0.000	0.000	0.001	0.000	0.000	0.001	0.000	0.001			-0.001

GAMS による応用一般均衡分析：資本市場と企業固有の生産要素

参 照 文 献

- 小平裕 (2002a) 「GAMS による応用一般均衡分析：基本モデル」『経済研究』156号, 2002年3月。
- 小平裕 (2002b) 「GAMS による応用一般均衡分析：中間投入のあるモデル」『経済研究』157号, 2002年5月。
- Brooke, A., Kendrick, D., and Meeraus, A., (1988), *GAMS: A User's Guide Release 2.25*, Boyd and Fraser.
- Lofgren, H., (1999), *Exercises in General Equilibrium Modeling using GAMS*, International Food Policy Research Institute.